/*
 * Copyright 1993-2009 NVIDIA Corporation.  All rights reserved.
 *
 * NVIDIA Corporation and its licensors retain all intellectual property and 
 * proprietary rights in and to this software and related documentation. 
 * Any use, reproduction, disclosure, or distribution of this software 
 * and related documentation without an express license agreement from
 * NVIDIA Corporation is strictly prohibited.
 *
 * Please refer to the applicable NVIDIA end user license agreement (EULA) 
 * associated with this source code for terms and conditions that govern 
 * your use of this NVIDIA software.
 * 
 */

/*
  Parallel reduction kernels
*/

#include <stdio.h>
#include "hb_utils.h"

#pragma comment(user, "CUDA NVCC Version " VERSION_MAJOR "." VERSION_MINOR)

static bool isPow2(unsigned int x)
{
    return ((x&(x-1))==0);
}

#ifdef __DEVICE_EMULATION__
#define EMUSYNC __syncthreads()
#else
#define EMUSYNC
#endif

// Utility class used to avoid linker errors with extern
// unsized shared memory arrays with templated type
template<class T>
struct SharedMemory
{
    __device__ inline operator       T*()
    {
        extern __shared__ int __smem[];
        return (T*)__smem;
    }

    __device__ inline operator const T*() const
    {
        extern __shared__ int __smem[];
        return (T*)__smem;
    }
};

// specialize for double to avoid unaligned memory 
// access compile errors
template<>
struct SharedMemory<double>
{
    __device__ inline operator       double*()
    {
        extern __shared__ double __smem_d[];
        return (double*)__smem_d;
    }

    __device__ inline operator const double*() const
    {
        extern __shared__ double __smem_d[];
        return (double*)__smem_d;
    }
};

#if 0
/*
  Parallel sum reduction using shared memory
  - takes log(n) steps for n input elements
  - uses n threads
  - only works for power-of-2 arrays
*/

/* This reduction interleaves which threads are active by using the modulo
   operator.  This operator is very expensive on GPUs, and the interleaved 
   inactivity means that no whole warps are active, which is also very 
   inefficient */
template <class T>
__global__ void
reduce0(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // load shared mem
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*blockDim.x + threadIdx.x;
    
    sdata[tid] = (i < n) ? g_idata[i] : 0;
    
    __syncthreads();

    // do reduction in shared mem
    for(unsigned int s=1; s < blockDim.x; s *= 2) {
        // modulo arithmetic is slow!
        if ((tid % (2*s)) == 0) {
            sdata[tid] += sdata[tid + s];
        }
        __syncthreads();
    }

    // write result for this block to global mem
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

/* This version uses contiguous threads, but its interleaved 
   addressing results in many shared memory bank conflicts. */
template <class T>
__global__ void
reduce1(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // load shared mem
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*blockDim.x + threadIdx.x;
    
    sdata[tid] = (i < n) ? g_idata[i] : 0;
    
    __syncthreads();

    // do reduction in shared mem
    for(unsigned int s=1; s < blockDim.x; s *= 2) 
        {
            int index = 2 * s * tid;

            if (index < blockDim.x) 
                {
                    sdata[index] += sdata[index + s];
                }
            __syncthreads();
        }

    // write result for this block to global mem
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

/*
  This version uses sequential addressing -- no divergence or bank conflicts.
*/
template <class T>
__global__ void
reduce2(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // load shared mem
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*blockDim.x + threadIdx.x;
    
    sdata[tid] = (i < n) ? g_idata[i] : 0;
    
    __syncthreads();

    // do reduction in shared mem
    for(unsigned int s=blockDim.x/2; s>0; s>>=1) 
        {
            if (tid < s) 
                {
                    sdata[tid] += sdata[tid + s];
                }
            __syncthreads();
        }

    // write result for this block to global mem
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

/*
  This version uses n/2 threads --
  it performs the first level of reduction when reading from global memory
*/
template <class T>
__global__ void
reduce3(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // perform first level of reduction,
    // reading from global memory, writing to shared memory
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*(blockDim.x*2) + threadIdx.x;

    sdata[tid] = (i < n) ? g_idata[i] : 0;
    if (i + blockDim.x < n) 
        sdata[tid] += g_idata[i+blockDim.x];  

    __syncthreads();

    // do reduction in shared mem
    for(unsigned int s=blockDim.x/2; s>0; s>>=1) 
        {
            if (tid < s) 
                {
                    sdata[tid] += sdata[tid + s];
                }
            __syncthreads();
        }

    // write result for this block to global mem 
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

/*
  This version unrolls the last warp to avoid synchronization where it 
  isn't needed
*/
template <class T, unsigned int blockSize>
__global__ void
reduce4(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // perform first level of reduction,
    // reading from global memory, writing to shared memory
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*(blockDim.x*2) + threadIdx.x;

    sdata[tid] = (i < n) ? g_idata[i] : 0;
    if (i + blockSize < n) 
        sdata[tid] += g_idata[i+blockSize];  

    __syncthreads();

    // do reduction in shared mem
    for(unsigned int s=blockDim.x/2; s>32; s>>=1) 
        {
            if (tid < s)
                {
                    sdata[tid] += sdata[tid + s];
                }
            __syncthreads();
        }

#ifndef __DEVICE_EMULATION__
    if (tid < 32)
#endif
        {
            if (blockSize >=  64) { sdata[tid] += sdata[tid + 32]; EMUSYNC; }
            if (blockSize >=  32) { sdata[tid] += sdata[tid + 16]; EMUSYNC; }
            if (blockSize >=  16) { sdata[tid] += sdata[tid +  8]; EMUSYNC; }
            if (blockSize >=   8) { sdata[tid] += sdata[tid +  4]; EMUSYNC; }
            if (blockSize >=   4) { sdata[tid] += sdata[tid +  2]; EMUSYNC; }
            if (blockSize >=   2) { sdata[tid] += sdata[tid +  1]; EMUSYNC; }
        }

    // write result for this block to global mem 
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}
#endif

/*
  This version is completely unrolled.  It uses a template parameter to achieve 
  optimal code for any (power of 2) number of threads.  This requires a switch 
  statement in the host code to handle all the different thread block sizes at 
  compile time.
*/
template <class T, unsigned int blockSize>
__global__ void
reduce5(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // perform first level of reduction,
    // reading from global memory, writing to shared memory
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*(blockSize*2) + threadIdx.x;

    sdata[tid] = (i < n) ? g_idata[i] : 0;
    if (i + blockSize < n) 
        sdata[tid] += g_idata[i+blockSize];  

    __syncthreads();

    // do reduction in shared mem
    if (blockSize >= 512) { if (tid < 256) { sdata[tid] += sdata[tid + 256]; } __syncthreads(); }
    if (blockSize >= 256) { if (tid < 128) { sdata[tid] += sdata[tid + 128]; } __syncthreads(); }
    if (blockSize >= 128) { if (tid <  64) { sdata[tid] += sdata[tid +  64]; } __syncthreads(); }
    
#ifndef __DEVICE_EMULATION__
    if (tid < 32)
#endif
        {
            if (blockSize >=  64) { sdata[tid] += sdata[tid + 32]; EMUSYNC; }
            if (blockSize >=  32) { sdata[tid] += sdata[tid + 16]; EMUSYNC; }
            if (blockSize >=  16) { sdata[tid] += sdata[tid +  8]; EMUSYNC; }
            if (blockSize >=   8) { sdata[tid] += sdata[tid +  4]; EMUSYNC; }
            if (blockSize >=   4) { sdata[tid] += sdata[tid +  2]; EMUSYNC; }
            if (blockSize >=   2) { sdata[tid] += sdata[tid +  1]; EMUSYNC; }
        }
    
    // write result for this block to global mem 
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

/*
  This version adds multiple elements per thread sequentially.  This reduces the overall
  cost of the algorithm while keeping the work complexity O(n) and the step complexity O(log n).
  (Brent's Theorem optimization)
*/
template <class T, unsigned int blockSize, bool nIsPow2>
__global__ void
reduce6(T *g_idata, T *g_odata, unsigned int n)
{
    T *sdata = SharedMemory<T>();

    // perform first level of reduction,
    // reading from global memory, writing to shared memory
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*(blockSize*2) + threadIdx.x;
    unsigned int gridSize = blockSize*2*gridDim.x;
    sdata[tid] = 0;

    // we reduce multiple elements per thread.  The number is determined by the 
    // number of active thread blocks (via gridDim).  More blocks will result
    // in a larger gridSize and therefore fewer elements per thread
    while (i < n) {
        sdata[tid] += g_idata[i];
        // ensure we don't read out of bounds -- this is optimized away for powerOf2 sized arrays
        if (nIsPow2 || i + blockSize < n) 
            sdata[tid] += g_idata[i+blockSize];  
        i += gridSize;
    } 
    __syncthreads();

    // do reduction in shared mem
    if (blockSize >= 512) { if (tid < 256) { sdata[tid] += sdata[tid + 256]; } __syncthreads(); }
    if (blockSize >= 256) { if (tid < 128) { sdata[tid] += sdata[tid + 128]; } __syncthreads(); }
    if (blockSize >= 128) { if (tid <  64) { sdata[tid] += sdata[tid +  64]; } __syncthreads(); }
    
#ifndef __DEVICE_EMULATION__
    if (tid < 32)
#endif
        {
            if (blockSize >=  64) { sdata[tid] += sdata[tid + 32]; EMUSYNC; }
            if (blockSize >=  32) { sdata[tid] += sdata[tid + 16]; EMUSYNC; }
            if (blockSize >=  16) { sdata[tid] += sdata[tid +  8]; EMUSYNC; }
            if (blockSize >=   8) { sdata[tid] += sdata[tid +  4]; EMUSYNC; }
            if (blockSize >=   4) { sdata[tid] += sdata[tid +  2]; EMUSYNC; }
            if (blockSize >=   2) { sdata[tid] += sdata[tid +  1]; EMUSYNC; }
        }
    
    // write result for this block to global mem 
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

////////////////////////////////////////////////////////////////////////////////
// Wrapper function for kernel launch
////////////////////////////////////////////////////////////////////////////////
template <class T>
void 
reduce(int size, int threads, int blocks, 
       int whichKernel, T *d_idata, T *d_odata)
{
    dim3 dimBlock(threads, 1, 1);
    dim3 dimGrid(blocks, 1, 1);
    int smemSize = threads * sizeof(T);

    // choose which of the optimized versions of reduction to launch
    switch (whichKernel) {
#if 0
    case 0:
        reduce0<T><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size);
        break;
    case 1:
        reduce1<T><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size);
        break;
    case 2:
        reduce2<T><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size);
        break;
    case 3:
        reduce3<T><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size);
        break;
    case 4:
        switch (threads) {
        case 512:
            reduce4<T, 512><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 256:
            reduce4<T, 256><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 128:
            reduce4<T, 128><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 64:
            reduce4<T,  64><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 32:
            reduce4<T,  32><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 16:
            reduce4<T,  16><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  8:
            reduce4<T,   8><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  4:
            reduce4<T,   4><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  2:
            reduce4<T,   2><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  1:
            reduce4<T,   1><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        }
        break; 
#endif
    case 5:
        switch (threads) {
        case 512:
            reduce5<T, 512><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 256:
            reduce5<T, 256><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 128:
            reduce5<T, 128><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 64:
            reduce5<T,  64><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 32:
            reduce5<T,  32><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case 16:
            reduce5<T,  16><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  8:
            reduce5<T,   8><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  4:
            reduce5<T,   4><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  2:
            reduce5<T,   2><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        case  1:
            reduce5<T,   1><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
        }
        break;       
    case 6:
    default:
        if (isPow2(size)) {
            switch (threads) {
            case 512:
                reduce6<T, 512, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 256:
                reduce6<T, 256, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 128:
                reduce6<T, 128, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 64:
                reduce6<T,  64, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 32:
                reduce6<T,  32, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 16:
                reduce6<T,  16, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  8:
                reduce6<T,   8, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  4:
                reduce6<T,   4, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  2:
                reduce6<T,   2, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  1:
                reduce6<T,   1, true><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            }
        } else {
            switch (threads) {
            case 512:
                reduce6<T, 512, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 256:
                reduce6<T, 256, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 128:
                reduce6<T, 128, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 64:
                reduce6<T,  64, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 32:
                reduce6<T,  32, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case 16:
                reduce6<T,  16, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  8:
                reduce6<T,   8, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  4:
                reduce6<T,   4, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  2:
                reduce6<T,   2, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            case  1:
                reduce6<T,   1, false><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
            }
        }
        break;       
    }
}

//-----------------------------------------------------------------------------
// CuHB part
//-----------------------------------------------------------------------------
// Instantiate the reduction function for 3 types

template void 
reduce<int>(int size, int threads, int blocks, 
            int whichKernel, int *d_idata, int *d_odata);

template void 
reduce<float>(int size, int threads, int blocks, 
              int whichKernel, float *d_idata, float *d_odata);

template void 
reduce<double>(int size, int threads, int blocks, 
               int whichKernel, double *d_idata, double *d_odata);

//-----------------------------------------------------------------------------

void reduce_float_array(int size, int threads, int blocks, float *d_idata, float *d_odata)
{
    reduce<float>(size, threads, blocks, 5, d_idata, d_odata);
}

//-----------------------------------------------------------------------------

__device__ static inline float4 operator+(float4& a, float4 &b)
{
    return make_float4(a.x+b.x, a.y+b.y, a.z+b.z, a.w+b.w);
}

//-----------------------------------------------------------------------------

template <unsigned int blockSize>
__global__ void
reduce5_float4(float4 *g_idata, float4 *g_odata, unsigned int n)
{
    float4 *sdata = SharedMemory<float4>();
    float4 float4_zero = make_float4(0,0,0,0);
    // perform first level of reduction,
    // reading from global memory, writing to shared memory
    unsigned int tid = threadIdx.x;
    unsigned int i = blockIdx.x*(blockSize*2) + threadIdx.x;
    
    sdata[tid] = (i < n) ? g_idata[i] : float4_zero;
    if (i + blockSize < n) 
        sdata[tid] = sdata[tid] + g_idata[i+blockSize];  

    __syncthreads();

    // do reduction in shared mem
    if (blockSize >= 512) { if (tid < 256) { sdata[tid] = sdata[tid] + sdata[tid + 256]; } __syncthreads(); }
    if (blockSize >= 256) { if (tid < 128) { sdata[tid] = sdata[tid] + sdata[tid + 128]; } __syncthreads(); }
    if (blockSize >= 128) { if (tid <  64) { sdata[tid] = sdata[tid] + sdata[tid +  64]; } __syncthreads(); }
    
#ifndef __DEVICE_EMULATION__
    if (tid < 32)
#endif
        {
            if (blockSize >=  64) { sdata[tid] = sdata[tid] + sdata[tid + 32]; EMUSYNC; }
            if (blockSize >=  32) { sdata[tid] = sdata[tid] + sdata[tid + 16]; EMUSYNC; }
            if (blockSize >=  16) { sdata[tid] = sdata[tid] + sdata[tid +  8]; EMUSYNC; }
            if (blockSize >=   8) { sdata[tid] = sdata[tid] + sdata[tid +  4]; EMUSYNC; }
            if (blockSize >=   4) { sdata[tid] = sdata[tid] + sdata[tid +  2]; EMUSYNC; }
            if (blockSize >=   2) { sdata[tid] = sdata[tid] + sdata[tid +  1]; EMUSYNC; }
        }
    
    // write result for this block to global mem 
    if (tid == 0) g_odata[blockIdx.x] = sdata[0];
}

//-----------------------------------------------------------------------------

void reduce_float4_array(int size, int threads, int blocks, float4 *d_idata, float4 *d_odata)
{
    //reduce<float4>(size, threads, blocks, whichKernel, d_idata, d_odata);

    dim3 dimBlock(threads, 1, 1);
    dim3 dimGrid(blocks, 1, 1);
    int smemSize = threads * sizeof(float4);

    switch (threads) {
    case 512:
        reduce5_float4< 512><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case 256:
    //     reduce5<float4, 256><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case 128:
    //     reduce5<float4, 128><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case 64:
    //     reduce5<float4,  64><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case 32:
    //     reduce5<float4,  32><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case 16:
    //     reduce5<float4,  16><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case  8:
    //     reduce5<float4,   8><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case  4:
    //     reduce5<float4,   4><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case  2:
    //     reduce5<float4,   2><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    // case  1:
    //     reduce5<float4,   1><<< dimGrid, dimBlock, smemSize >>>(d_idata, d_odata, size); break;
    default:
        error("invalid threads number\n"); break;
    }
}

//-----------------------------------------------------------------------------
//-----------------------------------------------------------------------------

/*
 * Local variables:
 *  mode: c++
 *  c-indent-level: 4
 *  c-basic-offset: 4
 *  tab-width: 4
 *  indent-tabs-mode: nil
 * End:
 */
